

CONCOURS: LES MEILLEURS LOGICIELS

Vous concevrez vous-même vos propres programmes. Vous êtes disposé à nous les soumettre et à participer à notre concours « Les meilleurs logiciels ».

Vos programmes devront résoudre un ou plusieurs calculs en vue d'une utilisation pratique (gestion, jeux, comptabilités, routines astucieuses, etc.), avec un listing ne dépassant pas 60 lignes.

Les conditions de participation à ce concours seront publiées dans le numéro 2 de MICRO 7 (à paraître le 25 janvier 1983). Vous pouvez d'ores et déjà nous écrire :

MICRO 7, Edi 7, 6, rue Ancelle, 92525 Neuilly cedex.


Casse-tête d'un cheminot

(lire notre article p. 97)

Un exemple de logiciel à réaliser chez soi... Piquez-vous au jeu. Élaborez un programme et envoyez-le-nous. Respectez le passage de train dans l'ordre décrit sur le schéma p. 98, en sachant que : l'ordinateur n'intervient que sur les aiguillages.

Fixez-vous arbitrairement une distance entre les trains, en fonction de la limite des cantons que vous déterminerez. La solution vous sera donnée dans le numéro 2 de MICRO 7.

Hypothèse de départ :
le train A va en W,
B et C vont en X,
D va en W
(1^{er} trajet à tester)



© Copyright MICRO 7 ou auteurs, Paris décembre 1982. Tous droits réservés. Ont contribué à ce cahier : Denis Bolland, Michel D. Brivot, Pascal Chauvin, M. Mazet, Roger Muller, Olivier Rouillard.

LE CAHIER DU LOGICIEL

Chaque numéro de MICRO 7 vous propose un éventail de programmes originaux à saisir directement sur votre micro-ordinateur. Nous avons rendu chacun de ces programmes adaptables au plus grand nombre de machines. Lorsque cela n'a pas été possible, nous vous précisons par quelles modifications il convient de remplacer les instructions trop spécifiques. Tous les domaines d'applications seront régulièrement abordés : gestion, comptabilité, éducation, jeux...

NOTRE COTATION

très facile •
facile ••
moyen •••
difficile ••••
très difficile •••••

Les fiches du mois :

GESTION
Intérêts capitalisés
Salaire brut ou net
Amortissements linéaires dégressifs

VIE PRATIQUE

Lever du soleil
Calendrier
Tiercé

JEUX

Loto
Poker
Mot masqué
© Copyright Micro 7, auteurs, décembre 82.

APPLE, PET CBM, TRS
80 etc.
Mémoire minimale : 16 K
Langage : BASIC
Difficulté : **

INTERETS CAPITALISES

Pierre qui roule amasse mousse, ou des économies qui font des petits; telle pourrait être la devise du petit écureuil de la Caisse d'épargne !

Le calcul des intérêts capitalisés est une chose aisée si vous avez à votre disposition une bonne calculette ou une table financière. Il est parfois pratique de faire ressortir tous les éléments (capital, intérêts capitalisés et intérêts totaux) afin de cerner aux mieux les flux économiques.

L'idée de base est qu'après une période déterminée arbitrairement, les intérêts s'incorporent au capital. A l'époque suivante, le capital ainsi augmenté donnera des intérêts plus élevés.

C'est comme si à la 2^e période l'on verse des intérêts sur les anciens intérêts.

Ex. : Capital de départ K
Intérêts annuels I
Taux d'intérêt T
Intérêts totaux TI

Le programme ici présenté, décompose, par période de 4 mois, les mouvements financiers. Il fait apparaître le capital après chaque période, le total des intérêts de l'année et le total de ces mêmes intérêts sur plusieurs années.

L'étape première initialise le programme où "SI" est le total des intérêts reçus, "P" le nombre de périodes dans l'année (ici 3) et "TA" symbolise la tabulation.

Puis, l'on introduit les premiers paramètres (N : nombres d'années; K : le capital et T : le taux d'intérêt).

Un test met T automatiquement sous la forme d'un pourcentage.

La séquence suivante calcule l'intérêt de la période, capitalise les intérêts et formate l'écriture. En effet les chiffres doivent être alignés à partir de la droite. Les heureux possesseurs d'un "PRINT USING" n'auront pas toutes ces difficultés mais tout le monde n'a pas la chance de disposer d'un Basic étendu.

Ce sous-programme de formatage effectue aussi l'arrondi au franc. Rien ne vous empêche, pour être plus précis, de calculer le capital et intérêt au centime près. Il transforme ce chiffre en chaîne de caractères puis en fonction de la longueur, imprime à l'emplacement désiré.


L'opération se répète également pour l'affichage des intérêts versés dans l'année.

Vous pouvez, pour cerner de plus près la réalité, noter la ligne :

76 LET T = T/P

où « T » représente le taux d'intérêt annuel et « P » le nombre de périodes dans l'année.

Cette technique de simulation de "PRINT USING" n'est certainement pas la meilleure, mais elle me semble assez facile à comprendre, si l'on s'en donne la peine, c'est pourquoi je vous la propose. D'autres méthodes sont les bienvenues, alors... à vous d'améliorer le système.



LISTING DU PROGRAMME:

```

10 LET P=3
30 LET SI=NOT P
40 INPUT N
50 INPUT K
60 INPUT T
70 IF T>1 THEN LET T=T/100
72 PRINT "CAPITAL ";K;" TAUX ";T;" EN ";N;" ANS"
75 PRINT "P    K    K    K    I"
76 LET T=T/P
80 FOR J=1 TO N
82 LET S=0
85 LET TA=1
100 PRINT J;
110 FOR U=1 TO P
115 LET I=K*T
120 LET S=S+I
125 LET SI=SI+I
130 LET K=K+I
135 LET A=K
140 GOSUB 500
160 NEXT U
170 PRINT "=>";
180 LET A=S
185 LET TA=TA+2
190 GOSUB 500
205 PRINT
210 NEXT J
215 PRINT
220 PRINT "TOTAL ";INT(SI*100)/100
250 STOP
500 IF A<>INT A THEN LET A=A+.5
510 LET A$=STR$(INT A)
515 LET TA=TA+7
520 PRINT TAB(TA-LEN A$);A$;
530 RETURN
    
```

SINCLAIR ZX-80
Mémoire minimale : 1K
Langage : BASIC
Difficulté : **

LOTO



Chaque mercredi soir, assis devant votre poste, vous rêvez de châteaux en Espagne. Voici un programme pour augmenter vos chances de gain au Loto.

À défaut de donner une martingale infaillible, procurant richesse et bonheur, ce programme de Loto vous familiarisera avec les jeux de hasard.

Le principe fondamental de cet aléatoire est de ne pas tirer deux fois la même boule.

Il faut donc conserver la valeur du nombre « X » et vérifier qu'il ne figure pas dans la liste établie par les précédents tirages.

Plus le nombre de boules à « pêcher » est grand, plus votre micro-ordinateur mettra de temps.

Voici quelques explications à l'aide de statistiques très simples :

Vous pouvez vous-mêmes faire l'expérience en tirant les 49 boules. Pour cela, il faudra néanmoins modifier le programme en mettant :

```

20 DIM T (50)
100 FOR I = 1 TO 49
    
```

et en effaçant la ligne 150. A l'exécution, vous remarquerez que l'affichage, rapide au début, met de plus en plus de temps à s'effectuer. De ce petit logiciel, une autre application est possible.

	Probabilité d'avoir déjà été tiré	Probabilité de ne pas avoir déjà été tiré
1 ^{re} boule	0	49/49
2 ^e boule	1/49	48/49
3 ^e boule	2/49	47/49
» » »	» » »	» » » »
» » »	» » »	» » » »
» » »	» » »	» » » »
48 ^e boule	47/49	2/49
49 ^e boule	48/49	1/49

Je vous pose le problème :

« Combien de parties dois-je jouer, avant de gagner? »

Pour cela vous devrez entrer votre grille et faire « tourner » le Loto en comptant les essais jusqu'à ce que vous gagniez...

Répétez l'opération plusieurs fois pour faire apparaître une moyenne significative proche de l'estimation statistique.

Pour le moment, comme vous dit si bien l'ordinateur : « à vos grilles et bonne chance ! »...

LISTING DU PROGRAMME:

```

10 RAND
20 DIM T (8)
30 LET T (1) = 0
40 PRINT "      LOTO"
50 PRINT
60 PRINT
70 PRINT
80 PRINT "  LES 6 CHIFFRES PROPOSÉS"
90 PRINT "  SONT : ",.....
100 FOR I = 1 TO 7
110 LET X = INT (RND * 49) + 1
120 FOR J = 1 TO I
130 IF X = T (J) THEN GOTO 110
140 NEXT J
150 IF I = 7 THEN PRINT,..... "LE NUMÉRO COMPLÉ-
MENTAIRE";
160 PRINT " "; X; " ";
170 LET T (I) = X
180 NEXT I
190 PRINT
200 PRINT
210 PRINT "  BONNE CHANCE..."
220 STOP
    
```




APPLE, PET CBM, TRS
80 etc.
Mémoire minimale : 1K
Langage : BASIC
Difficulté : *

POKER

Paire... Double paire... Breelan... Full aux As! le cigare à la bouche, le revolver à portée de main : une partie en « franche amitié » avec votre compagnon de jeu...

Il s'agit d'une variante du très célèbre poker. Le programme a été très simplifié pour tenir sur 1 Ko RAM mais vous pouvez toujours le compléter. Il se joue ici avec 9 cartes, sans distinction de couleur. Après le premier tirage vous avez 2 possibilités. Conserver les cartes où les changer. Tapez 0 jusqu'à ce que

le numéro de la carte à retirer apparaisse, puis n'importe quoi et sa valeur s'effacera pour une nouvelle donne. Au deuxième tirage, l'ordinateur comptabilisera vos points en Livres Sterling. Rien ne vous empêche de transformer ces points en centimes. Attention : l'ordinateur a les nerfs solides et surtout rien à perdre.



LISTING DU PROGRAMME :

```

22 LET E=5
23 LET G=E/E
24 DIM B$(E)
26 LET M=E*E
28 PAUSE E*E*E
30 CLS
100 FOR I=G TO E
101 LET V=0
106 LET B$(I)=STR$ INT (RND*9+G)
130 NEXT I
135 GOSUB 900
160 FOR I=G TO E
165 PRINT AT 10,10; I
168 PAUSE 4E4
170 IF INKEY$="0" THEN GOTO 175
172 LET B$(I)=STR$ INT (RND*9+G)
173 PRINT AT E+G,I*E+G;" "
175 NEXT I
176 CLS
180 GOSUB 900
190 FOR I=G TO 9
195 LET N=0
200 FOR J=G TO E
210 IF VAL B$(J)=I THEN LET N=N+G
220 NEXT J
230 IF N<=G THEN GOTO 300
240 LET V=V+N*(G+N)
300 NEXT I
310 LET V=V-E-E
320 LET M=M+V
330 PRINT " ";V,M;"£"
340 GOTO 100
900 LET F=E
950 FOR I=G TO E
1000 PRINT AT E,F;" "
1010 PRINT AT E+G,F;" ";B$(I);" "
1020 PRINT AT E+G+,F;" "
1021 LET F=F+E
1025 NEXT I
1030 RETURN
    
```

SHARP PC 1211
Mémoire minimale : 2K
Langage : BASIC
Difficulté : **

CALENDRIER

**La date de mon mariage? Facile! Quel jour au juste?
Ça devait être un samedi... non, un vendredi.
Je me souviens qu'il pleuvait, à moins que ce ne soit la veille.
Fera-t-on le pont pour le 1^{er} Mai 1984?**

Voici un petit programme pour micro-ordinateur de poche du type SHARP PC 1211, qui délivre le jour de la semaine, à partir de sa date, ou un calendrier complet portant sur un mois entier. Vous pourrez ainsi retrouver le jour de votre date de naissance, ou prévoir vos dates de vacances de l'année prochaine ou savoir quand tombe Noël... Son utilisation est des plus simples. Après avoir introduit le programme (fig. 1), l'ordinateur vous posera quelques questions concernant le jour, le mois et l'année désirés, auxquelles

vous devrez répondre avec précision. Imaginons que vous vouliez connaître le jour de la semaine correspondant au 7/6/82. Il vous suffira de répondre par 7, 6 et 1982 aux questions « jour? », « mois? », et « année? ». Le résultat s'affiche ensuite immédiatement (fig. 2). Pour obtenir le mois complet, rien de plus simple. Il suffit d'entrer 0 (zéro) à la question « jour? ». La figure 3 présente le résultat d'une telle opération pour le mois de février 1920. En février de cette

FÉVRIER 1920	
1 DIMANCHE	16 LUNDI
2 LUNDI	17 MARDI
3 MARDI	18 MERCREDI
4 MERCREDI	19 JEUDI
5 JEUDI	20 VENDREDI
6 VENDREDI	21 SAMEDI
7 SAMEDI	22 DIMANCHE
8 DIMANCHE	23 LUNDI
9 LUNDI	24 MARDI
10 MARDI	25 MERCREDI
11 MERCREDI	26 JEUDI
12 JEUDI	27 VENDREDI
13 VENDREDI	28 SAMEDI
14 SAMEDI	29 DIMANCHE
15 DIMANCHE	

LISTING DU PROGRAMME :

```

1:"JOUR1" CLEAR
2:REM "R.M
02/04/198 2
10:INPUT "JOUR
?";J: IF J<>0
LET F=1:GOTO
20
11:J=1
20:INPUT "MOIS
? ";M
23:IF (M>12)+(M
<1)GOTO 20
25:INPUT "ANNEE
?";A
26:IF (A<1582)+
(A>2000)GOTO
25
32:USING "
":GOSUB 600
33:GOSUB 500+M
34:GOSUB 600
35:USING " "
40:X=INT (A/100
)
50:W=A-(X*100)
60:IF (M=1)+(M=
2)GOTO 90
80:N=0:GOTO 140
90:IF W=0GOTO 1
15
100:B=W-((INT (W
/4))*4)
105:IF B=0LET N=
1:GOTO 140
110:N=2:GOTO 140
115:D=X-((INT (X
/4))*4)
120:IF D=0LET N=
1:GOTO 140
130:N=2
140:IF M<>2GOTO
147
142:IF (N=0)+(N=
2)LET Y=28:
GOTO 147
144:Y=29
147:C=INT (365.2
5*W)+INT (30
.56*M)+J+N
150:S=C/7:S=INT
S:S=C-(S*7):
S=S+200:GOTO
S
200:PRINT "I";J;
"MERCREDI
I":GOTO 400
201:PRINT "I";J;
"JEUDI
I":GOTO 400
202:PRINT "I";J;
"VENDREDI
I":GOTO 400
203:PRINT "I";J;
"SAMEDI
I":GOTO 400
204:PRINT "I";J;
"DIMANCHE
I":GOTO 400
205:PRINT "I";J;
"LUNDI
I":GOTO 400
206:PRINT "I";J;
"MARDI
I":GOTO 400
400:IF F=1END
401:IF S=206LET
S=199
402:S=S+1
409:J=J+1
410:IF (J=Y+1)+(
S=204)+(S=20
5)PRINT "I--
-----
I"
411:IF (J=Y+1)
END
420:GOTO S
501:PRINT " JANV
IER ";A:Y=3
1:RETURN
502:PRINT " FEVR
IER ";A:
RETURN
503:PRINT " MA
RS ";A:Y=3
1:RETURN
504:PRINT " AVR
IL ";A:Y=3
0:RETURN
505:PRINT " MA
I ";A:Y=3
1:RETURN
506:PRINT " JU
IN ";A:Y=3
0:RETURN
507:PRINT " JUI
LLET ";A:Y=3
1:RETURN
508:PRINT " AO
UT ";A:Y=3
1:RETURN
509:PRINT " SEPT
EMBRE";A:Y=3
0:RETURN
510:PRINT " OCT
OBRE ";A:Y=3
1:RETURN
511:PRINT " NOVE
MBRE ";A:Y=3
0:RETURN
512:PRINT " DECE
MBRE ";A:Y=3
1:RETURN
540:IF M=13LET M
=1:A=A+1
555:J=1:GOTO 32
600:PRINT "-----
":RETURN
    
```

SINCLAIR ZX-81
Mémoire minimale : 16 K
Langage : BASIC
Difficulté : ***

AMORTISSEMENT

Certaines règles comptables sont obligatoires dans une société. La comptabilisation des amortissements en fait partie. Elle a de plus l'avantage de s'inspirer de la réalité.

Vous achetez un bien, une machine par exemple. Avec le temps, votre machine s'use, s'abîme. Et cela jusqu'au moment où elle ne vaudra plus rien. Si vous avez eu la bonne idée de faire des économies, vous pourrez acheter une autre machine. Par exemple, une machine de 10 000 F sera inutilisable dans 10 ans.

Il suffit donc de mettre chaque année pleine 10 000/10 soit 1 000 F de côté. Cela constitue l'amortissement linéaire.

Vous pourrez au besoin décider d'amortir plus au début puis moins par la suite. On appelle cela le dégressif. On multipliera alors l'amortissement linéaire par un coefficient variable, suivant la durée de vie du bien.

Durée de vie	Coefficient multiplicateur
2 à 4 ans	1,5
4 à 6 ans	2

plus de 6 ans 2,5

Ainsi pour un matériel dont la durée normale d'utilisation est de 5 ans, le taux d'amortissement dégressif est de $20\% \times 2 = 40\%$. De toutes façons les amortissements seront déduits du bénéfice fiscal soumis à l'impôt.

Comme le minimum de l'amortissement toléré est égal au linéaire, le système dégressif diminuera jusqu'à ce seuil pour redevenir linéaire.

Si la date d'achat de l'immobilisation ne correspond pas à la date d'ouverture du bilan, il suffira de faire une règle de trois (supposant l'année à 360 jours).

Cette technique appelé « prorata temporis » est calculée au jour le jour pour l'amortissement linéaire et mois par mois dans le cas dégressif.

L'administration exige chaque année qu'un relevé des amortissements soit joint à la déclaration des revenus.

Notons, pour être précis, que ce programme abandonne les chiffres derrière la virgule. Cela entraîne une petite erreur mais vous pouvez toujours améliorer...

Variables :
A : nombre d'années
T : durée de vie
P : prix des immobilisations
C : amortissements
D : cumul des amortissements
A\$: date
B\$: variable d'entrée

VOULEZ-VOUS

- 1) LES AMORTISSEMENTS LINÉAIRES
- 2) LES AMORTISSEMENTS DÉGRESSIFS
- 3) LES DEUX

ANNÉE	AMORT. LINÉAIRE	V.N.C.
80	2466	117534
81	8000	109534
82	8000	101534
83	8000	93534
84	8000	85534
85	8000	77534
86	8000	69534
87	8000	61534
88	8000	53534
89	8000	45534
90	8000	37534
91	8000	29534
92	8000	21534
93	8000	13534
94	8000	5534
95	5534	0

IMMO DE 120 000 FRANCS
AMORTIS EN 15 ANS, TAUX DE 6 %

ANNÉE	AMORT. DÉGRESSIF	V.N.C.
80	6666	113334
81	18889	94445
82	15740	78705
83	13117	65588
84	10931	54657
85	9109	45548
86	7591	37957
87	6326	31631
88	5271	26360
89	4393	21967
90	4393	17574
91	4393	13181
92	4393	8788
93	4393	4395
94	4393	2

IMMO DE 120 000 FRANCS
AMORTIS EN 15 ANS, TAUX DE 6 %

VOUS POUVEZ :

- 1) CHANGER DE VARIABLES
- 2) REFAIRE LES MÊMES CALCULS
- 3) ARRÊTER

```

100 DATA "SEENGOLPYUPCAEA"
110 DATA "ETAFNNIAPICUCND"
120 DATA "RUOGINACONCAGUA"
130 DATA "EAANDESPSROCREV"
140 DATA "VANETONAGOLYMPE"
150 DATA "EREBUSINAISERAN"
200 REM CLEAR1000
210 L=15: C=15
220 DIM A$(L), T$(L, C)
230 FOR I=1 TO L: READ A$(I): NEXT I
240 FOR I=1 TO L
250 FOR J=1 TO C: T$(I, J)=MID$(A$(I), J, 1)
260 NEXT J, I
270 INPUT "MOT À CHERCHER"; X1$
280 D$=LEFT$(X1$, 1): F$=RIGHT$(X1$, 1):
X=LEN(X1$): X1=X-1: X$=""
290 FOR I=1 TO X: X$=MID$(X1$, I, 1)+X$: NEXT I
300 FOR I=1 TO L
310 FOR J=1 TO C: IF D$=T$(I, J) THEN 350
320 NEXT J, I
330 PRINT "IL Y A UNE ERREUR DANS LA GRILLE OU
DANS LE MOT"
340 END
350 T1=L-I-X1: T2=I-X1: T3=C-J-X1: T4=J-X1
360 S1=I+X1: S2=J+X1: D1=I-X1: D2=J-X1
370 IF T1<0 THEN 390
380 IF F$=T$(S1, J) THEN 540
390 IF T2<0 THEN 410
400 IF F$=T$(D1, J) THEN 570
410 IF T3<0 THEN 430
420 IF F$=T$(I, S2) THEN 600
430 IF T4<0 THEN 450
440 IF F$=T$(I, D2) THEN 630
450 IF T1<0 OR T3<0 THEN 470
460 IF F$=T$(S1, S2) THEN 660
470 IF T2<0 OR T4<0 THEN 490
480 IF F$=T$(D1, D2) THEN 690
490 IF T1<0 OR T4<0 THEN 510
500 IF F$=T$(S1, D2) THEN 720
510 IF T2<0 OR T3<0 THEN 320
520 IF F$=T$(D1, S2) THEN 750
530 GOTO 320
540 Y$=""
550 FOR K=0 TO X1: Y$=T$(I+K, J)+Y$: NEXT K
560 IF Y$=X$ THEN 780 ELSE 390
570 Y$=""
580 FOR K=0 TO X1: Y$=T$(I-K, J)+Y$: NEXT K
590 IF Y$=X$ THEN 780 ELSE 410
600 Y$=""
610 FOR K=0 TO X1: Y$=T$(I, J+K)+Y$: NEXT K
620 IF Y$=X$ THEN 780 ELSE 430
630 Y$=""
640 FOR K=0 TO X1: Y$=T$(I, J-K)+Y$: NEXT K
650 IF Y$=X$ THEN 780 ELSE 450
660 Y$=""
670 FOR K=0 TO X1: Y$=T$(I+K, J+K)+Y$: NEXT K
680 IF Y$=X$ THEN 780 ELSE 470
690 Y$=""
700 FOR K=0 TO X1: Y$=T$(I-K, J-K)+Y$: NEXT K
710 IF Y$=X$ THEN 780 ELSE 490
720 Y$=""
730 FOR K=0 TO X1: Y$=T$(I+K, J-K)+Y$: NEXT K
740 IF Y$=X$ THEN 780 ELSE 510
750 Y$=""
760 FOR K=0 TO X1: Y$=T$(I-K, J+K)+Y$: NEXT K
770 IF Y$=X$ THEN 780 ELSE 320
780 PRINT "COORDONNÉES I="; I; "J="; J
790 GOTO 270
    
```

LE MOT MYSTÉRIEUX

par Roger La Ferté

Tous les mots se retrouvant dans cette grille peuvent y figurer en tous sens : horizontalement, verticalement, en diagonale, de haut en bas et vice-versa, de droite à gauche et inversement. Les mots de la grille se croisent, et donc une lettre peut servir pour plusieurs mots, sauf celles du mot mystère. Le dernier à trouver ne figurant pas dans la liste est le mot mystère défini comme suit : **Massif montagneux.**

AAR	ELVEND	PARNASSE
ACONCAGUA	EREBUS	PIC
ADRAR	ESPEROU	PINDE
ALPES	ESTEREL	PUY
ANDES	ETNA	PYRÉNÉES
ANETO	EVEREST	RHODOPE
ANNAPURNA	FOUTA-DJALON	RHON
CALPE	FUJI-YAMA	RIGHI
CANIGOU	HARZ	SANCY
CASSIN	ISERAN	SINAI
CAUCASE	LOGAN	TOURMALET
CERVINE	NEBO	VERCORS
CHAINE	NEVADA	VISO
CHIMBORAZO	OLYMPE	VOLCAN
DAHRA	OSSA	VOSGES
DONON	OURAL	
EIFEL	PANGEE	

Aucune lettre morte dans cette grille

MOT À CHERCHER : ACONCAGUA
COORDONNÉES I = 12 J = 7

MOT À CHERCHER : ALPES
COORDONNÉES I = 6 J = 8

MOT À CHERCHER : ANDES
COORDONNÉES I = 13 J = 3

MOT À CHERCHER : ANNAPURNA
COORDONNÉES I = 3 J = 11

MOT À CHERCHER : ALBA
IL Y A UNE ERREUR DANS LA GRILLE OU DANS LE MOT

L'utilisation du programme.

	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
1	D	A	H	R	A	R	D	A	M	A	Y	I	J	U	F
2	N	O	H	R	C	H	I	M	B	O	R	A	Z	O	E
3	E	T	N	A	V	O	S	G	E	S	A	E	U	R	S
4	V	O	N	O	S	D	U	Z	H	N	T	N	E	S	
5	L	U	I	I	N	O	E	R	N	I	A	A	E	P	A
6	E	R	V	S	E	P	L	A	A	D	C	N	N	S	N
7	R	M	R	I	L	E	P	H	J	L	E	E	I	E	R
8	E	A	E	A	A	U	C	A	O	R	A	S	S	O	A
9	T	L	C	E	R	T	L	V	Y	C	N	A	S	B	P
10	S	E	E	N	G	O	L	P	Y	U	P	C	A	E	A
11	E	T	A	F	N	N	I	A	P	I	C	U	C	N	D
12	R	U	O	G	I	N	A	C	O	N	C	A	G	U	A
13	E	A	A	N	D	E	S	P	S	R	O	C	R	E	V
14	V	A	N	E	T	O	N	A	G	O	L	Y	M	P	E
15	E	R	E	B	U	S	I	N	A	I	S	E	R	A	N

Un exemple de grille de mot masqué (doc. Télé 7 jours).

Matériel : Tout micro disposant d'instructions
READ, DATA et IF. THEN. ELSE
Langage : BASIC
Difficulté : ***

LE MOT MYSTERIEUX

Cachés dans une grille de caractères, des mots échappent à la vue du joueur. Le camouflage est un art difficile, mais ceux-là semblent le pratiquer avec talent. Ce programme écrit en Basic se chargera, à votre place, de trouver les mots dissimulés.

Les règles du jeu sont simples et font surtout appel à votre mémoire visuelle. On raye dans les huit directions (verticale haut ou bas, horizontale

leurs, voici sur un exemple emprunté au grand cruciverbiste Roger La Ferté, comment fonctionne ce jeu. L'ordinateur, mettra entre « 3 et 20 secondes » pour



gauche ou droite, diagonale en haut droite ou gauche, diagonale en bas droite ou gauche) les mots à insérer dans la grille jusqu'au dernier. Alors apparaît comme par enchantement le mot astucieusement camouflé qu'il fallait trouver. D'ail-

vous indiquer sans faillir les coordonnées I, J de ces mots que vous recherchez depuis un quart d'heure.

Ce programme Basic, dont le listing est donné, ci-après, fait surtout appel à des ordres de traitement et d'identification de chaînes

de caractères. A l'aspect ludique s'ajoute donc un côté pédagogique sur l'utilisation qu'on peut faire des ordres ésotériques tels que LEN (X1\$), MID\$(A\$(I), J, I).

Afin de s'y retrouver dans le comportement de ce logiciel, voici une description de sa structure où l'on peut constater que la manipulation des chaînes de caractères est l'une de ses principales activités.

10-150 : En tête de programme, la grille est introduite grâce à l'instruction DATA.

200-260 : Introduction des données dans les tableaux A\$ et T\$. A\$ contient directement les lignes de la grille sous forme de chaînes de caractères alors que T\$ est l'image directe de cette grille. Aux coordonnées I, J, se trouve le caractère correspondant. La ligne 200 n'est valable que pour le micro-ordinateur TRS-80 et doit être omise sur tous les autres.

270-340 : Routine principale de recherche d'un mot dans la grille. La première et la dernière lettre du mot à retrouver sont

placées dans les variables D\$ et F\$. La première servira à détecter le début d'un mot dans la grille et la dernière à éviter une recherche systématique de tous les caractères composant ce mot.

350-530 : En utilisant la première et la dernière lettre du mot, la possibilité de trouver ce mot est examinée dans les 8 directions pré-citées. Si les lettres correspondent, un branchement est effectué à une séquence de recherche systématique de toutes les lettres du mot.

540-770 : Recherche systématique de toutes les lettres appartenant à un mot, dans les 8 directions.

Si le mot n'est pas trouvé, (Y\$ différent de X\$) une recherche est effectuée dans une autre direction. Signalons, pour les Basic ne disposant pas de l'instruction ELSE, qu'il est possible de transformer l'écriture du programme. Ainsi la ligne 560 peut s'écrire de la façon suivante :

560 IF X\$=Y\$ THEN 780
565 GOTO 390

780 : Affichage à l'écran des coordonnées du mot.

LISTING DU PROGRAMME:

```
10 DATA "DAHRARDAMAYIJUF"
20 DATA "NOHRCHIMBORAZOE"
30 DATA "ETNAVOSGESAEURS"
40 DATA "VONOSDUZHNTNES"
50 DATA "LUIINOERNIAAEP"
60 DATA "ERVSEPLAADCNNSN"
70 DATA "RMRILEPHJLEIEER"
80 DATA "EAEAAUCAORASSOA"
90 DATA "TLCERTLVYCNASBP"
```

LISTING DU PROGRAMME:

```
3 CLS
4 REM AMORTISSEMENTS
5 REM
6 REM
7 REM AUTEUR : PASCAL CHAUVIN
8 REM
9 REM
10 PRINT "-----"
12 PRINT ":", TAB 31, ":"
13 PRINT ":", TAB 31, ":"
14 PRINT " AMORTISSEMENTS"; TAB 31, ":"
15 PRINT ":", TAB 31, ":"
16 PRINT "LINEAIRES OU DEGRESSIFS"; TAB 31, ":"
17 PRINT ":", TAB 31, ":"
18 PRINT ":", TAB 31, ":"
19 PRINT "-----"
20 PRINT
90 LET X=0
100 REM ENTREES DES DONNEES
101 REM
110 PRINT "PRIX DES IMMOBILISATIONS ";
120 INPUT P
130 PRINT P
150 PRINT "DUREE DE VIE ";
160 INPUT N
170 PRINT N
180 IF N>5 THEN FAST
200 PRINT "DATE D" "ACHAT JJ.MM.AA"
210 INPUT A$
230 IF LEN A$ <> 8 THEN GOTO 210
240 IF VAL A$(4 TO 5) > 12 THEN GOTO 210
250 IF VAL A$(1 TO 2) > 31 THEN GOTO 210
260 PRINT AT 12,13; A$
262 PRINT "VOULEZ VOUS CHANGER? O/N"
263 INPUT B$
264 IF B$ = "O" OR B$ = "0" THEN GOTO 3
PRINT "....." "VOULEZ VOUS", "1) LES AMORTISSEMENTS LINEAIRES"; "2) LES AMORTISSEMENTS DEGRESSIFS 3) LES DEUX"
267 LET X=0
268 INPUT Z
270 IF Z=1 THEN GOTO 276
271 IF Z=2 THEN GOTO 281
272 GOTO 290
276 LET X=1
277 GOTO 290
281 LET E=1000
282 LET B$="DEGRESSIF"
283 CLS
284 GOTO 305
290 LET E=0
295 CLS
299 REM
300 REM TABLEAU AMORT
301 REM
302 LET B$="LINEAIRE"
305 LET D=0
310 PRINT " "
320 PRINT " ANNEE AMORT "; B$; " V.N.C. "
330 PRINT " "
350 LET A=VAL A$(7 TO 8)
400 GOSUB 8000+E
410 PRINT " "
420 LET B$="DEGRESSIF"
440 PRINT "IMMO DE "; P; " FRANCS", "AMORTIS EN "; N; " ANS, TAUX DE "; INT (100/N); " "
445 IF X=1 THEN GOTO 9800
450 INPUT C$
460 CLS
470 GOTO 305
```

```
7999 REM
8000 REM AMORT LINEAIRE
8010 LET B=1-(VAL A$(4 TO 5)-1)*30-1+VAL A$(1 TO 2)/360
8020 FOR I=1 TO N
8030 LET C=INT (P*B/N)
8040 LET D=D+C
8050 PRINT " "; A; TAB (15-LEN STR$ C); C; TAB (30-LEN STR$ (P-D)); P-D; TAB 31; " "
8060 LET A=A+1
8080 LET B=1
8090 NEXT I
8100 PRINT " "; A; TAB (15-LEN STR$ (P-D)); P-D; TAB (29); "0 "
8110 LET E=1000
8120 RETURN
8999 REM
9000 REM AMORT DEGRESSIF
9001 REM
9010 REM COEFFICIENT
9020 IF N<=2 THEN GOTO 9600
9030 IF N<=4 THEN LET K=1,5
9040 IF N<=6 THEN LET K=2
9050 IF N>6 THEN LET K=2,5
9100 REM INITIALISATION
9101 REM
9110 LET A=VAL A$(7 TO 8)
9120 LET B=1-(VAL A$(4 TO 5)-1)/12
9130 LET D=0
9140 LET F=P
9200 REM CALCUL
9201 REM
9205 LET C=INT (F*B*K/N)
9206 LET B=1
9210 FOR I=1 TO N
9220 LET D=D+C
9230 PRINT " "; A; TAB (15-LEN STR$ C); C; TAB (30-LEN STR$ (P-D)); P-D; TAB 31; " "
9260 LET B=1
9270 LET A=A+1
9280 LET F=F-C
9285 LET C=INT (F*B*K/N)
9290 IF C<=(P-D)/(N-I) THEN GOTO 9400
9300 NEXT I
9400 REM ON REVIENT AU LINEAIRE
9401 REM
9410 LET C=INT (F/(N-I))
9420 FOR J=1 TO N-I
9430 PRINT " "; A; TAB (15-LEN STR$ C); C; TAB (30-LEN STR$ (F-C)); F-C; TAB 31; " "
9450 LET A=A+1
9455 LET F=F-C
9460 NEXT J
9465 LET X=1
9500 RETURN
9600 PRINT "..... IMMOBILISATION DE MOINS DE 2 ANS : PAS D'AMORTISSEMENT DEGRESSIF ...."
9798 REM MENU
9799 REM
9800 INPUT Z$
9802 CLS
9805 SLOW
9810 PRINT "VOUS POUVEZ : "
9812 PRINT "1) CHANGER DE VARIABLES"
9814 PRINT "2) REFAIRE LES MEMES CALCULS"
9816 PRINT "3) ARRETER"
9818 INPUT I
9819 IF N>5 THEN FAST
9820 IF I=1 THEN GOTO 90
9830 IF I=3 THEN STOP
9900 GOTO 266
```

APPLE, PET CBM, TRS 80 etc.
Mémoire minimale : 16K
Langage : Basic
Difficulté : **

TIERCE

Joueurs de tiercé, laissez faire les chiffres. Ou plutôt l'ordinateur. Voici un exemple d'exécution de notre programme Spécial tiercé, à jouer sur votre micro.

Il s'agit de faire mathématiquement le meilleur pronostic à partir de vos pronostiqueurs habituels. Nous ne prenons pas de pourcentage sur les gains...

Exemple d'exécution

NOMBRE DE CHEVAUX	15
COMBIEN DE PRONOSTICS	5
ENREGISTREMENT	
PRONOSTIC NUMÉRO	1

DONNEZ LES NUMÉROS DES CINQ PREMIERS CHEVAUX.	3 8 7 12 15
QUELLE EST LA CONFIANCE QUE VOUS ACCORDEZ À CE PRONOSTIC (NOTE DE 1 À 10)	5
PRONOSTIC NUMÉRO	2
DONNEZ LES NUMÉROS DES CINQ PREMIERS CHEVAUX.	12 3 11 14 4
QUELLE EST LA CONFIANCE QUE VOUS ACCORDEZ À CE PRONOSTIC (NOTE DE 1 À 10)	3
PRONOSTIC NUMÉRO	3
DONNEZ LES NUMÉROS DES CINQ PREMIERS CHEVAUX.	14 12 5 1 3
QUELLE EST LA CONFIANCE QUE VOUS ACCORDEZ À CE PRONOSTIC (NOTE DE 1 À 10)	7
PRONOSTIC NUMÉRO	4
DONNEZ LES NUMÉROS DES CINQ PREMIERS CHEVAUX.	7 12 6 3 2
QUELLE EST LA CONFIANCE QUE VOUS ACCORDEZ À CE PRONOSTIC (NOTE DE 1 À 10)	8
PRONOSTIC NUMÉRO	5
DONNEZ LES NUMÉROS DES CINQ PREMIERS CHEVAUX.	11 12 3 7 9
QUELLE EST LA CONFIANCE QUE VOUS ACCORDEZ À CE PRONOSTIC (NOTE DE 1 À 10)	2

VOICI MON PRONOSTIC

VOUS DEVEZ CHOISIR LES CHEVAUX DONT LA NOTE EST LA PLUS ÉLEVÉE

CHEVAL NUMÉRO 1 NOTE = 14
CHEVAL NUMÉRO 2 NOTE = 8
CHEVAL NUMÉRO 3 NOTE = 66

CHEVAL NUMÉRO 4 NOTE = 3
CHEVAL NUMÉRO 5 NOTE = 21
CHEVAL NUMÉRO 6 NOTE = 24
CHEVAL NUMÉRO 7 NOTE = 59
CHEVAL NUMÉRO 8 NOTE = 20
CHEVAL NUMÉRO 9 NOTE = 2
CHEVAL NUMÉRO 10 NOTE = 0
CHEVAL NUMÉRO 11 NOTE = 19
CHEVAL NUMÉRO 12 NOTE = 93
CHEVAL NUMÉRO 13 NOTE = 0
CHEVAL NUMÉRO 14 NOTE = 41
CHEVAL NUMÉRO 15 NOTE = 5

BONNE CHANCE!

LISTING DU PROGRAMME:

```

10 REM COPYRIGHT CLAIRE GOUARD ET DENIS BO-
LAND
20 REM ASSOCIATION ORMES
30 REM PROGRAMME ESSAYE SUR COMMODORE ET
APPLE
40 REM POUR JOUER AU TIERCÉ OU AU QUARTÉ
50 INPUT "NOMBRE DE CHEVAUX ";C
60 PRINT
70 INPUT "COMBIEN DE PRONOSTICS ";P
80 DIMS(C)
90 PRINT
100 PRINT "ENREGISTREMENT:"
110 FOR I=1 TO P
120 PRINT "PRONOSTIC NUMÉRO ";I
130 PRINT
140 PRINT "DONNEZ LES NUMÉROS DES CINQ"
150 INPUT "PREMIERS CHEVAUX. ";C1,C2,C3,C4,C5
160 PRINT
170 PRINT "QUELLE EST LA CONFIANCE QUE"
180 PRINT "VOUS ACCORDEZ À CE PRONOSTIC"
190 INPUT "(NOTE DE 1 À 10)";CF
200 S(C1)=S(C1)+CF*5
210 S(C2)=S(C2)+CF*4
220 S(C3)=S(C3)+CF*3
230 S(C4)=S(C4)+CF*2
240 S(C5)=S(C5)+CF*1
250 NEXT I
260 PRINT
270 PRINT "VOICI MON PRONOSTIC:"
280 PRINT
290 PRINT "VOUS DEVEZ CHOISIR LES CHEVAUX"
300 PRINT "DONT LA NOTE EST LA PLUS ÉLEVÉE:"
310 PRINT
320 FOR I=1 TO C
330 PRINT "CHEVAL NUMÉRO ";I;"NOTE=";S(I)
340 PRINT "APPUYEZ SUR 'C' POUR CONTINUER"
350 GETC :IFC$<>"C"THEN350
360 PRINT
370 NEXT I
380 PRINT:PRINT "BONNE CHANCE!"

```

Puis en mode Réserve :

RÉSERVE 1

F1 : GOTO 20 Init (initialisation)
F2 : GOTO 120 Lev
F3 : GOTO 140 Cou
F4 : GOTO 160 Mer
F5 : GOTO 170 H, Z

GUIDE DES INTRODUCTIONS (PC 1500 seule)

INTRODUCTIONS	APPUYER	AFFICHAGE
Au début de tout NOUVEAU CALCUL :		
RESERVE I....	Init	Date?
Date : JJMM.AAAA...	ENTER	Latitude?
Latit. : DD°MM' (-si Sud)	ENTER	Longitude?
Long. : DDD°MM' (-si Est)	ENTER	RCL I : Quel Problème?
Si le Problème est : Lever : RESERVE I...	Lev ENTER	Z = DDD.d, Hre = HH.MM ; ENTER Deb AUBE : HH.MM : RCL I?
Si le Problème est : Coucher : RESERVE I...	Cou ENTER	Z = DDD.d, Hre = HH.MM ; ENTER Fin de CREPUSC. : HH.MM. : RCL I?
Si le Problème est : Hre Méridienne : RESERVE I...	Mer	Hre TU MER. : HH.MM : RCL I?
Si le Problème est : Coordonnées Horiz. RESERVE I	H, Z	Hre des coord. H,Z?
Heure des coord. Horiz.	ENTER	Z = DDD.d, Hr = DD.MM. : RCL I?

TABEAU DES VARIABLES

N° de Ligne	Variable	Affectation de la VARIABLE
20	D	: Date Introduite au format : JJMM.YYYY
	J	: Jour
	M	: Mois
30	Y	: Année
40	A	: Partie entière de Y/100
	T	: Temps en fraction décimale de siècle depuis 1900,0
50	L	: Latitude, introduite au format : D, m (degrés, minutes)
	G	: Longitude, introduite au format : D, m (degrés, minutes)
75	EM	: Équation du temps moyen
80	I	: Variable de boucle
	H	: Heure méridienne
	S	: Longitude du Soleil
	M	: Anomalie moyenne
90	K	: Équation du centre
100	Q	: Déclinaison du Soleil
110	P	: Angle au Pôle au lever ou au coucher
	D	: corr. à P Pôle au lever ou au coucher
	AB	: Hautr. du Soleil en « dessous » de l'horizon (en degrés)
	X	: Azimut au lever ou au coucher
	W	: corr. à X « »
170	U	: Hre désirée des coord. horizontales
	J	: Angle Horaire à l'Hre U
180	I	: Hauteur en degrés du Soleil au-dessus de l'horizon
	Y	: numérateur de : SINJ / (COSJ × SINL - TANQ × COSL)
	D	: dénominateur de :
	M	: Module : racine carrée de Y ² + D ²
190	Z	: Azimut du Soleil à l'Hre désirée des coord. horiz.

EXEMPLE DE CALCUL À ROUEN

DATE : 16.10.1982

Latit. : 49.30
Long. : - 1.00

LEVER :
Azimut = 103.3
Heure = 6.20
Deb AUBE : 5.47

COUCHER :
Azimut = 256.6
Heure = 17.02
Fin de CRÉPUSCULE : 17.35

Hre TU MÉRIDienne : 11.41

Coord. Horizon. a : 9.45
Hres TU
Azimut = 147.7
Hauteur = 25.57

Coord. Horizon. a : 11.40
Hres TU
Azimut = 179.8
Hauteur = 31.14

EXEMPLE DE CALCUL PACIFIQUE NORD

DATE : 31.10.1979

Latit. : 45.00
Long. : - 179.55


LEVER :
Azimut = 109.2
Heure = - 5.22
Deb AUBE : - 5.54

COUCHER :
Azimut = 250.7
Heure = 4.50
Fin de CRÉPUSCULE : 5.21

Hre TU MÉRIDienne : - 0.16

Coord. Horizon. a :
- 1.30 Hres TU
Azimut = 159.4
Hauteur = 28.34

Coord. Horizon. a :
- 2.30 Hres TU
Azimut = 144.2
Hauteur = 23.33



SHARP PC 1 500
Mémoire minimale : 6 K
Langage : BASIC
Difficulté : ***

LEVER DU SOLEIL

LISTING DU PROGRAMME :

Entrer en mode Program les instructions suivantes :

LEVER COUCHER DU SOLEIL

```

4 : USING "      " : RETURN
6 : USING "      " : RETURN
20 : INPUT "DATE TU ? :"; D : J=INT (D / 100), M=INT
      (D-J*100)
30 : Y=(D-INT D)*10^4 : IF M <=2 LET Y = Y - 1,
      M=M+12
40 : A=INT (Y/100), T=(INT (365.25*Y)+INT (30.6*
      (M+1))-A+INT (A/4)+J-694023.5)/36525
50 : INPUT "Latitude ? :"; L : L=DEGL : INPUT "Longi-
      tude ? :"; G : G=DEG G
75 : EM=0
80 : FOR I=0 TO 1 : H=180+G-EM :
      T=T+H/13149000 : S=36000.77*T-80.31,
      M=35999.05*T-1.524
90 : K=SIN M*(1.919-.005*T)+SIN (2*M)*.02,
      S=S+K : EM=2.47*SIN (2*S)-1.91*SIN M : NEXT I
100 : Q=ASN .3978*SIN S
105 : PRINT "RCL 1 : QUEL PROBLEME?"
110 : P=ACS (-TAN L*TAN Q) : D=AB/√((COS L)^
      2-(SIN Q)^2) : W=D*SIN L : X=ACS (SIN Q/COS
      L) : RETURN

120 : AB=.8 : GOSUB 110 : GOSUB 6 : PRINT "Z=";
      X-W; USING "      " : Hre :";
      DMS ((H-P-D)/15); " : ENTER" : AB=6
130 : GOSUB 110 : PRINT "Deb AUBE :"; DMS
      ((H-P-D)/15); " : RCL 1?"

140 : AB=.8 : GOSUB 110 : GOSUB 6 : PRINT "Z=";
      360+W-X; USING "      " : Hre :";
      DMS ((H+P+D)/15); " : ENTER"
150 : AB=6 : GOSUB 110 : PRINT "Fin de CREP. :";
      DMS ((H+P+D)/15); " : RCL 1?"

160 : GOSUB 4 : PRINT "Hre TU MER. :"; DMS (H/15);
      " : RCL 1?"

170 : INPUT "Hre TU des coord. H, Z? :"; U : U=DEG
      U : J=U*15+180+EM-G
180 : I=ASN (COS J*COS Q*COS L+SIN L*SIN Q),
      Y=SIN J, D=COS J*SIN L-TAN Q*COS L,
      M=√(Y^2+D^2)
190 : IF Y<0 LET Z=-ACS (D/M)+360 : GOTO 210
200 : Z=ACS (D/M)
210 : Z=Z+180 : IF Z>=360 LET Z=Z-360
215 : GOSUB 6 : PRINT "Z="; Z;
      USING "      " : Hr=" : DMS I; " : RCL 1?"

```

Le soleil vous connaissez? Mais peut-être pas aussi bien que vous le supposez! En y regardant de plus près, on s'aperçoit que l'habitant de Strasbourg verra le soleil plus tôt que celui de Brest, quant au Dunkerquois il profitera, le même jour, moins longtemps des ardeurs solaires que le Niçois qui, de surcroît, le verra culminer plus haut!

Dans les heures de lever et de coucher du soleil, sa position dans le ciel dépend de votre position sur terre. Pour vous exercer, voici quelques coordonnées de villes qui feront ressortir ces différences :

N.B. TU = temps universel, ± 1H par rapport à Greenwich.

ROUEN : 1°00'E/49°30'N
BREST : 4°30'E/48°20'N
STRASBOURG : 7°50'E/
48°30'N
DUNKERQUE : 2°00'E/
51°00'N
PARIS : 0°20'E/48°50'N
PERPIGNAN : 3°00'E/
42°40'N
CAP TOWN : 18°00'E/
33°00'S
et... au Sud des Iles Aléou-
tiennes : 179°55'E/
45°00'N

Par ce programme, vous saurez tout! Et même plus, par exemple quelle sera la pièce la plus lumineuse et jusqu'où arrivera l'ombre en fonction des dates, des heures et l'inclinaison du soleil.



PET CBM
Mémoire minimale : 16K
Langage : Basic
Difficulté : **

NET A BRUT

FICHE DE PAIE PERSONNEL DE SERVICE			
DU 01/08/82 AU 31/08/82			
* Salaire de Base pour 169h.....			3319.16
* Salaire pour 3 jours supplémentaires.....			382.98
* 10 heures supplémentaires à 25%.....			245.50
* Conges payés.....			448.62
Avantages en nature :			
* 50 repas à 10.97.....			548.50
* 25 nuits à 7.08.....			177.00
SALAIRE BRUT....			5121.76

BASES	TAUX	RETENUES	
* 5121.76	5.60 %	SS maladie	286.82
* 5121.76	4.70 %	SS vieillesse	240.72
* 5121.76	1.76 %	Retraite comp	90.14
* 5121.76	0.84 %	ASSEDIC	43.02
TOTAL RETENUES			660.71
NET IMPOSABLE...			4461.05

Avantages en nature :			
* 50 repas à 10.97.....			548.50
* 25 nuits à 7.08.....			177.00
NET A PAYER....			3735.55

Certains employeurs garantissent à leurs salariés un salaire net de cotisations sociales. Il leur est donc nécessaire de reconstituer à partir du montant net un montant brut correspondant.

L'établissement logique d'une paye s'effectue en principe de la manière suivante :

- On calcule et regroupe tous les éléments qui constituent le salaire brut.
- On détermine les bases sur lesquelles s'effectuent les diverses retenues sociales.

- On calcule les retenues et on les soustrait du salaire brut pour obtenir le salaire net imposable.

- Éventuellement on ajoute les diverses indemnités non imposables (par exemple la prime de transport) et on déduit les acomptes etc.

Il existe pour cela des formules plus ou moins

complexes. En effet, toutes les retenues ne s'appliquent pas sur la même base et l'on doit tenir compte du plafond de la sécurité sociale et des éventuels abattements (figure 1).

En résumé et en schématisant, nous avons donc la suite logique suivante :

Brut

Brut - (brut ou partie de brut X divers taux de retenues) = Net imposable. D'où la nécessité évidente de connaître le salaire brut avant de pouvoir appliquer les retenues.

Rien de bien compliqué dans tout cela, mais la difficulté commence quand on ne connaît pas le salaire brut mais seulement le salaire net.

Pourquoi donc ne pas utiliser ce petit programme et laisser la machine effectuer les calculs, d'autant

Salaire net ou salaire brut? Voilà une question que l'on se pose toujours, que l'on soit à la recherche d'un nouvel emploi, ou que l'on ait à embaucher du personnel. Passer du salaire brut au salaire net est une opération aisée. Il suffit de déduire l'ensemble des retenues, calculées à partir de la somme initiale. Mais, pour obtenir le salaire brut en fonction du salaire net, la tâche se complique sérieusement. Afin de faciliter ce travail, nous vous proposons un petit programme appelé « NET A BRUT » qui effectue la reconstitution de salaires bruts à partir de leur montant initial.

plus que NET A BRUT peut être utilisé seul, ou adapté comme sous-programme général de paye.

NET A BRUT : un maniement très simple

L'utilisateur dispose d'un écran de saisie des données (fig. 2) sur lequel figurent tous les renseignements nécessaires à l'établissement de la paye.

À la mise en route de NET A BRUT l'ordinateur affiche les données inscrites dans le programme (ligne 170 à 230). Cela évite d'avoir à réintroduire à chaque fois tous les taux. Il sera donc nécessaire de modifier les lignes du programme lorsque le besoin s'en fera sentir.

Toutes les données figurant sur l'écran des saisies sont affectées d'un numéro. Pour modifier une donnée, il faut taper le numéro correspondant et donner ensuite le nouveau chiffre.

Dès que le montant net est entré, on peut taper sur la touche « = » pour avoir à l'écran des résultats (fig. 2).

Pour revenir à l'écran des saisies, on appuie sur une touche quelconque. Il est

ainsi possible d'introduire de nouvelles données.

Ce programme testé sur micro-ordinateur PET CBM, utilise seulement deux instructions particulières à cette machine :

PRINT CHR\$(147) qui permet l'effacement de l'écran.

POKE 158,0 qui vide le tampon clavier avant l'instruction GET.

En outre, NET A BRUT comprend une « routine » de formatage des chiffres à l'édition, bien utile pour les utilisateurs d'ordinateurs qui ne possèdent pas une instruction du type « printusing ». Cette routine force des zéros dans les blancs des centimes et permet ainsi un alignement correct de tous les chiffres par la virgule (ou plus exactement par le point).

L'organisation du programme

Ce programme dont le listing est donné figure 3 présente une structure très simple :

lignes 100 à 150 - Présentation
ligne 160 - Définition de la fonction d'arrondi
lignes 170 à 230 - Données conservées dans le

NET À BRUT

programme (facultatif car on peut les fournir sur l'écran des saisies) lignes 250 à 410 - Affichage de l'écran de saisie des données
 lignes 418 à 560 - Entrée du numéro de la ligne et du montant puis affectation et retour au nouvel écran des saisies. La ligne 440 teste la condition pour passer à la phase des calculs et ensuite à l'affichage des résultats.
 ligne 580 - Test préliminaire des calculs. Si le net fourni (N1) est supérieur au plafond SS (P) branchement direct aux calculs suivant mode 2
 lignes 590 à 700 - Calculs mode 1
 Ils correspondent aux calculs à effectuer pour un salaire brut inférieur au plafond. Les lignes 600 à 630 permettent de calculer le brut (B1). Si ce brut s'avère être supérieur au plafond (P), la ligne 640 renvoie au mode 2.
 Les lignes 650 à 700 déterminent le brut arrondi, la base brute, le brut plafonné, les différentes retenues, le total des retenues et le net après calculs (N2).
 lignes 710 et 720 - Branchement vers le sous-programme d'édition et retour
 lignes 740 à 860 - Calculs mode 2
 Ils correspondent aux calculs à effectuer pour un salaire brut supérieur au pla-

fond. (Même déroulement que pour le mode 1 avec ligne test en 800 et renvoi éventuel au mode 1)
 lignes 870 et 880 - Branchement vers le sous-programme d'édition et retour.
 ligne 900 - Sous-programme d'édition (écran des résultats)
 lignes 910 à 935 - Préparation de l'édition. Parfois le net obtenu après les calculs ne correspond pas au centime près, au net introduit (N1). En effet, le montant brut étant arrondi, il arrive qu'il y ait un centime d'écart au niveau du net dû aux brisées des retenues. Dans ce cas les instructions des lignes 910 à 935 rétablissent l'équilibre.
 lignes 940 à 1110 - Édition de l'écran des résultats avec appel au sous-programme de formatage placé en 1160 pour aligner les résultats d'une façon agréable.
 lignes 1115 à 1130 - Temporisation et retour, si l'on appuie sur une touche, à l'affichage de l'écran des saisies.
 De plus, il est possible d'adapter ce programme pour en faire un sous-programme dans un programme général de paye, en ne conservant que les lignes où s'effectuent les calculs et en accordant les variables de transmission des données avec le programme principal. ■

Salaire sous plafond :

$$\text{Brut} = \frac{\text{Salaire Net}}{1 - \left(\frac{T4}{100} + \left(\frac{T1+T2+T3}{100} \times \frac{100-A1}{100} \right) \right)}$$

Salaire supérieur au plafond :

$$\text{Brut} = \frac{\text{Salaire Net} + \left(\left(\frac{P \times 100 - A1}{100} \right) \times \frac{T2}{100} \right)}{1 - \left(\frac{T4}{100} + \left(\frac{T1+T3}{100} \times \frac{100-A1}{100} \right) \right)}$$

Figure 1 : Les formules utilisées pour calculer les salaires bruts à partir de leurs homologues nets...

LISTING DU PROGRAMME:

```

READY.
100 REM
110 REM ! NETABRUT
120 REM ! PROGRAMME DE RECONSTITUTION
130 REM ! DE SALAIRE - PET CBM
140 REM ! AUTEUR JOEL TARQUINJ
150 REM
160 DEF FNA(Z)=INT(Z*100+0.5)/100
170 REM TAUX
180 P=6590:REM PLAFOND SS
190 T1=5.50:REM TAUX SS S/TOTALITÉ
200 T2=4.80:REM - SS S/PLAFOND
210 T3=0.84:REM - CHOMAGE
220 T4=1.76:REM - RETRAITE
230 REM
240:REM NETTOYAGE DE L'ÉCRAN
250 PRINTCHR$(147);
260 PRINT" PROGRAMME 'NETABRUT'"
270 FORK=1TO39:PRINT"="::NEXTK
280 PRINT:PRINT:PRINT"DONNÉES :":PRINT
290 PRINT"1. PLAFOND SECURITÉ SOC :":P
300 PRINT"2. TAUX AS S/TOTALITÉ :":T1
310 PRINT"3. TAUX AS S/PLAFOND :":T2
320 PRINT"4. TAUX CHÔMAGE :":T3
330 PRINT"5. TAUX RETRAITE :":T4
340 PRINT
350 PRINT"6. SALAIRE NET :":N1
360 PRINT"7. ABATTEMENT EN % :":A1
370 FORK=1TO39:PRINT"="::NEXTK
380 PRINT:PRINT
390 PRINT"POUR FOURNIR UNE DONNÉE":PRINT
    TAPEZ -> LE NUMÉRO CORRESPONDANT"
    PRINT
400 PRINT"POUR EXÉCUTION TAPEZ SUR LA
    TOUCHE -> ="
418 POKE158,0
420 GET R$:IF R$="" THEN 420
430 R=VAL(R$)
440 IF R$="" AND N1>0 THEN 580
450 IF R>0 AND R<8 THEN 470
460 GOTO 420
470 PRINT:PRINT:PRINT R;
480 INPUT M
490 IF R=1 THEN P=M
500 IF R=2 THEN T1=M
510 IF R=3 THEN T2=M
520 IF R=4 THEN T3=M
530 IF R=5 THEN T4=M
540 IF R=6 THEN N1=M
550 IF R=7 THEN A1=M
560 GOTO 250
570:
580 IFN1=>P THEN740
590 REM - CALCULS MODE 1 -
600 X1=(100-A1)/100
610 T5=(T1+T2+T3)/100
620 T6=T4/100
630 B1=N1/(1-T6-(T5*X1))
640 IF B1*X1>P*X1 THEN 740
650 B2=FNA(B1):BB=FNA(B2*X1):BP=FNA(P*X1)
660 C1=FNA(BB*T1/100)
670 C2=FNA(BB*T2/100)
680 C3=FNA(BB*T3/100)
690 C4=FNA(B2*T4/100)
700 CT=C1+C2+C3+C4:N2=FNA(B1-CT)
710 GOSUB 900
720 GOTO 250
730:
740 REM - CALCULS MODE 2 -
750 X1=(100-A1)/100
760 T5=(T1+T3)/100

```

```

770 T6=T4/100
780 T7=T2/100
790 B1=(N1+((P*X1)*T7))/(1-T6-(T5*X1))
800 IF(N1/(1-T6-(T5+T7)*X1))*X1<P*X1 THEN 590
810 B2=FNA(B1):BB=FNA(B2*X1):BP=FNA(P*X1)
820 C1=FNA(BB*T1/100)
830 C2=FNA(BP*T2/100)
840 C3=FNA(BB*T3/100)
850 C4=FNA(B2*T4/100)
860 CT=C1+C2+C3+C4:N2=FNA(B1-CT)
870 GOSUB 900
880 GOTO 250
890:
900 REM - PRÉPARATION ÉDITION -
910 IF N2=N1 THEN 940
920 N2=N1:B2=N2+CT
930 IF A1=0 THEN BB=B2
935 IF A1=0 AND BP<P THEN BP=B2
940 REM - ÉDITION -
950 PRINT CHR$(147)
960 PRINT:PRINT
970 PRINT"NET : ";N1;" -> BRUT : ";B1
980 PRINT:PRINT:PRINT
990 X=B2:GOSUB 1160:PRINT" SALAIRE
    BRUT":TAB(32-L);"X$
1000 GOSUB 1140
1010 X=BB:GOSUB 1160:PRINT"BASE
    BRUTE":TAB(32-L)X$
1020 X=BP:GOSUB 1160:PRINT"BASE
    PLAFONNÉE":TAB(32-L)X$
1030 GOSUB 1140
1040 X=C1:GOSUB 1160:PRINT"AS /BASE
    BRUTE(";T1;"%"):TAB(32-L)X$
1050 X=C2:GOSUB 1160:PRINT"AS /BASE
    PLAF.(";T2;"%"):TAB(32-L)X$
1060 X=C3:GOSUB 1160:PRINT"CHÔMAGE
    (";T3;"%"):TAB(32-L)X$
1070 X=C4:GOSUB 1160:PRINT"RETRAITE
    (";T4;"%"):TAB(32-L)X$
1080 GOSUB 1140
1090 X=CT:GOSUB 1160:PRINT"TOTAL
    DES RETENUES":TAB(32-L)X$
1100 GOSUB 1140
1110 X=N2:GOSUB 1160:PRINT"NET
    IMPOSABLE":TAB(32-L)X$
1115 POKE158,0
1120 GET A$:IF A$="" THEN 1120
1130 RETURN
1140 PRINTTAB(24)"-----":RETURN
1150:
1160 REM - FORMATAGE -
1170 X$=STR$(X)
1180 IF MID$(X$,LEN(X$)-1,1)="" THEN
    X$=X$+"0":GOTO 1200
1190 IF X=INT(X) THEN X$=X$+".00"
1200 L=LEN(X$)
1210 RETURN
READY.
***

```

PROGRAMME 'NET A BRUT'	
DONNÉES :	
1. PLAFOND SÉCURITÉ SOC :	6590
2. TAUX AS S/TOTALITÉ :	5.5
3. TAUX AS S/PLAFOND :	4.8
4. TAUX CHÔMAGE :	0,84
5. TAUX RETRAITE :	1.76
6. SALAIRE NET :	7500
7. ABATTEMENT EN % :	0
POUR FOURNIR UNE DONNÉE TAPEZ -> LE NUMÉRO CORRESPONDANT POUR EXÉCUTION TAPEZ SUR LA TOUCHE -> =	

NET : 7500 -> BRUT : 8505.24483	
SALAIRE BRUT	8505.24
BASE BRUTE	8505.24
BASE PLAFONNÉE	6590.00
AS /BASE BRUTE (5.5 %)	467.79
AS /BASE PLAF. (4.8 %)	316.32
CHÔMAGE (0,84 %)	71.44
RETRAITE (1.76 %)	149.69
TOTAL DES RETENUES	1005.24
NET IMPOSABLE	7500.00

PROGRAMME 'NET A BRUT'	
DONNÉES :	
1. PLAFOND SÉCURITÉ SOC :	6590
2. TAUX AS S/TOTALITÉ :	5.5
3. TAUX AS S/PLAFOND :	4.8
4. TAUX CHÔMAGE :	0,84
5. TAUX RETRAITE :	1.76
6. SALAIRE NET :	7500
7. ABATTEMENT EN % :	10
POUR FOURNIR UNE DONNÉE TAPEZ -> LE NUMÉRO CORRESPONDANT POUR EXÉCUTION TAPEZ SUR LA TOUCHE -> =	

NET : 7500 -> BRUT : 8412.78665	
SALAIRE BRUT	8412.79
BASE BRUTE	7571.51
BASE PLAFONNÉE	5931.00
AS /BASE BRUTE (5.5 %)	416.43
AS /BASE PLAF. (4.8 %)	284.69
CHÔMAGE (0,84 %)	63.60
RETRAITE (1.76 %)	148.07
TOTAL DES RETENUES	912.79
NET IMPOSABLE	7500.00

Figure 2 : Écrans de saisie et résultats pour deux exemples d'utilisation du programme : Dans le premier exemple, nous avons introduit dans les données un salaire net de 7 500 F sans abattement (a). Le résultat ne s'est pas fait attendre (b). Dans le deuxième cas, nous avons entré un salaire supérieur au plafond (c). De plus l'abattement a été modifié afin de montrer la répercussion au niveau du résultat brut (d).

TABLEAU DES VARIABLES	
P	Plafond de la sécurité sociale
T1	Taux AS sur salaire total
T2	Taux AS sur salaire plafonné
T3	Taux chômage
T4	Taux retraite
N1	Salaire net fourni
N2	Salaire net obtenu après calculs
A1	Pourcentage d'abattement
B1	Salaire brut réel obtenu
B2	B1 arrondi
BB	Base brute soit salaire brut-abattement
BP	Base plafonnée soit plafond ss-abattement
C1	Montant des cotisations AS sur salaire total (base brute)
C2	Montant des cotisations AS sur salaire plafonné (base plafonnée)
C3	Montant des cotisations chômage
C4	Montant des cotisations retraite
CT	Total des retenues